
Excel-Übung “Stundenzettel” 

 

Problem: 

Die täglichen Arbeitsstunden von Mitarbeitern sollen für mehrere Monate und Jahre erfasst 

werden. 

 

Lösung: 

Damit die Datumsangaben, Wochentage und Mitarbeiternamen nicht immer wieder neu ein-

gegeben werden müssen, wird eine Lösung angestrebt, bei der diese Daten automatisch mit 

Hilfe von Formeln in die entsprechenden Zellen eingefügt werden. 

Der Dateiname wird folgendermaßen gebildet: Für den Mitarbeiter Karl-Heinz Meyer und das 

Jahr 2019 heißt die Datei:  

2019-Karl-Heinz Meyer-V190121.xlsx 

Der Dateiname besteht also aus: 

 Der vierstelligen Jahreszahl, 

 einem Minuszeichen, 

 dem Namen des Mitarbeiters mit einem Leerzeichen zwischen Vor- und Nachname, 

 einem Minuszeichen, 

 einer Versionsangabe (Buchstabe V und Jahr, Monat, Tag jeweils zweistellig) 

 der Dateinamenserweiterung xlsx. 

In dieser Datei gibt es für jeden Monat ein Blatt mit dem Namen „1-2019“ für den Januar, „2-

2019“ für den Februar, … usw. 

 

Hinweis: 

Die Versionsangabe im Dateinamen „2019-Vorname Nachname-V190121.xlsx“ muss nicht 

geändert werden, wenn Daten (Arbeitszeiten) eingegeben werden – sondern nur, wenn das 

Layout oder die Funktionalität der Excel-Blätter geändert werden! 

 

Noch ein Hinweis: 

Wie sollen Sie diese Übung benutzen? Alle durchzuführenden Schritte sind so detailliert be-

schrieben, dass es auch einem völlig Ungeübten nicht schwerfällt, zum Ziel zu kommen. Es 

geht aber nicht nur darum, dass Sie alles hübsch „nachklicken“, was hier beschrieben wird. 

Sie sollen dabei beispielhaft die Benutzung der entsprechenden Excel-Techniken lernen, um 

sie dann auch in anderen Zusammenhängen anwenden zu können! Diese Techniken sind im 

Text rot markiert und am Ende nochmal in einer Liste zusammengefasst. 

Es wird erwartet, dass Sie diese Techniken anschließend beherrschen!! 

Der Name des Arbeitnehmers in D4 wird automatisch aus dem Namen der Datei entnommen. 

Der Name des Monats in I4 wird automatisch aus dem Blattnamen entnommen. 

Die Wochentage und Datumsangaben in den Spalten A und B werden automatisch für den 

aktuellen Monat (aus I4) gebildet. 

Der Benutzer kann Eintragungen in den Spalten C, D, E, G, I, J und K machen. Alles andere 

wird mit Formeln berechnet. 

Die Werte in den Spalten I und J braucht der Benutzer nicht einzutippen. Er kann sie mit Hilfe 

entsprechender Comboboxen eingeben. 

Bitte probieren Sie das zunächst einmal in der fertigen Datei aus, bevor Sie daran gehen, 
diese Datei selber zu erstellen!  

Anleitung für die Erstellung eines Blattes: 

1  
Machen Sie sich mit der Funktionalität der fertigen Excel-Datei vertraut: 

 Tragen Sie Arbeitszeiten, Pausen und Fahrzeiten ein und beobachten Sie die 

Reaktion in Zeile 40! 

 Benutzen Sie die Kombinationsfelder in den Spalten I und J, um dort Werte 

einzutragen! (U = Urlaub, K = Krank, F = frei, FT = Feiertag) 

 Versuchen Sie, in den Spalten I und J andere als die dort in den Kombinations-

feldern jeweils zugelassenen Werte einzutragen! 

 Beobachten Sie die Reaktion in Zeile 40 auf ein X in Spalte I! (Eine Erläute-

rung dazu kommt in Pkt. 30) 

 Legen Sie ein neues Blatt für einen weiteren Monat an (Rechtsklick auf den 

Blattnamen und Verschieben oder Kopieren wählen), benennen Sie es um 

(Rechtsklick auf den Blattnamen und Umbenennen wählen) und beobachten 

Sie die Reaktion in den Spalten A und B. Haben Sie registriert, dass die Zeilen 

mit Samstagen und Sonntagen automatisch gefärbt wurden?  

2  
Legen Sie eine neue Excel-Datei an, speichern Sie sie unter dem Namen 

“2019-Vorname Nachname-VJJMMTT.xlsx“ (V = Version). Dabei ist JJMMTT das ak-

tuelle Datum – bestehend aus jeweils zwei Stellen für das Jahr, den Monat und den 

Tag. 

Benennen Sie das Blatt „Tabelle1“ um in „1-2019“ (Rechtsklick auf den Blattnamen 

und Umbenennen wählen) . Stellen Sie die Seitenränder oben und unten auf 0,5cm 

ein: 

Seitenlayout / Seite einrichten / Seitenränder / Benutzerdefinierte Seitenränder 

Stellen Sie die Seitenansicht ein: 

Ansicht / Arbeitsmappenansichten / Seitenlayout 

3  
Markieren Sie das gesamte Blatt durch einen Klick ganz links oben (siehe Abbildung) 

 

und stellen Sie als Schriftart Arial 11 ein: Start / Schriftart 

4  
Verbinden Sie die Zellen A1 bis J1 (Start / Ausrichtung / Verbinden und zentrieren) 

und schreiben Sie hinein: 

„Arbeitnehmerbezogene Erfassung der täglichen Arbeitszeit“ 

Schriftart: Arial 12 fett unterstrichen 

5  
Verbinden Sie die Zellen A3 bis C3 und formatieren Sie sie auf Linksbündig / Oben: 

Schreiben Sie hinein: „Arbeitgeber:“ 

Schriftart: Arial 11 fett 

ACHTUNG: Wenn Sie Zellen zusammenfassen und diese zusammengefasste Zelle 

später in einer Formel benutzen wollen, müssen Sie sie mit der linken oberen Ecke 

adressieren. Die drei zusammengefassten Zellen A3, B3 und C3 mit dem Wort „Ar-

beitgeber“ darin haben also die Adresse A3! 

6  
Verbinden Sie die Zellen A4 bis C4 und formatieren Sie sie auf Linksbündig / Oben. 

Schreiben Sie hinein: „Arbeitnehmer:“ 

Schriftart: Arial 11 fett 

7  
Verbinden Sie die Zellen D3 bis F3 und formatieren Sie sie auf Linksbündig / Oben. 

Schreiben Sie den Namen des Arbeitgebers hinein – z.B. „Baufix GmbH“ 

Schriftart: Arial 10 unterstrichen 

8  
Verbinden Sie die Zellen D4 bis F4 und formatieren Sie sie auf Linksbündig / Oben. 

Schreiben Sie vorerst nichts hinein. 

Schriftart: Arial 10 unterstrichen 

9  
Schreiben Sie in Zelle G4 „Monat:“ 

Schriftart: Arial 11 fett 

10  
Verbinden Sie die Zellen H4 bis I4 und formatieren Sie sie auf Linksbündig / Oben. 

Schreiben Sie vorerst nichts hinein. 

11  
Schreiben Sie in die Zellen A6 bis J6 die Texte entspr. der Abbildung auf Seite 2 die-

ser Anleitung hinein. Wählen Sie als Umrandung Alle Rahmenlinien: 

ACHTUNG: Einen Zeilenumbruch innerhalb einer Zelle erreichen Sie mit Alt-Enter 

statt nur mit Enter! 

12  
Markieren Sie den Bereich A8:J38 und wählen Sie als Umrandung ebenfalls Alle Rah-

menlinien. Dann sieht Ihr Blatt jetzt so aus: 

 

Beachten Sie, dass einige Spalten rechtsbündig, einige linksbündig und einige mittig 

ausgerichtet sind! 

13  
Jetzt wird’s etwas kompliziert. Wir wollen den Namen des Arbeitnehmers und den Mo-

nat nicht selber eintippen, sondern automatisch aus dem Dateinamen bzw. dem Blatt-

namen entnehmen. Dafür tippen Sie folgende Formel in Zelle A5 ein: 

=ZELLE("dateiname";$A$1) 

Mit "dateiname" ist wirklich gemeint, dass Sie "dateiname" schreiben – nicht den 

Dateinamen. 

Daraufhin erscheint in A5 z.B. folgendes: (Bei Ihnen sieht das natürlich etwas anders 

aus, weil Sie die Datei in einem anderen Ordner als ich gespeichert haben!) 

D:\Lehre\Office\Excel\Stundenzettel\[2019-Vorname Nachname.xlsx]1-2019 

Das ist zunächst der Ordner, in dem sich die Datei befindet: 

D:\Lehre\Office\Excel\Stundenzettel\ 

es folgt der Dateiname in eckigen Klammern: 

[2019-Vorname Nachname.xlsx] 

und schließlich der Blattname: 

1-2019 

Das Ergebnis der o.g. Formel ZELLE(…) enthält also alles, was wir benötigen: Den 

Arbeitnehmernamen (im Dateinamen) und den Monat (im Blattnamen). 

Wir müssen also das, was wir brauchen, nur noch aus dem Inhalt von Zelle A5 „her-

ausschneiden“. 

ACHTUNG: Die Sache mit den Dollarzeichen in der Zelladresse $A$1 wird in Punkt 22 

erläutert. 

14  
Schreiben Sie die Formel 

=TEIL(A5;FINDEN("]";A5)+1;31) 

in Zelle A2. Sie schneidet einen Teil (darum heißt sie ja auch TEIL ;-) aus A5 heraus 

– und zwar beginnend bei Position FINDEN("]";A5)+1 (d.h. rechts neben der 

schließenden eckigen Klammer) und von dort aus die nächsten 31 Zeichen. 

Die Formel FINDEN(…) findet also die Position der schließenden eckigen Klammer in 

dem Text in Zelle A5. 

ACHTUNG: Die Formel FINDEN(…) liefert also als Ergebnis eine Zahl – keinen Text! 

In unserem Beispiel wäre das die 64 – bei Ihnen wird das natürlich eine andere Zahl 

sein, da Sie Ihre Datei in einem anderen Ordner gespeichert haben. 

Warum werden 31 Zeichen herausgeschnitten? Weil das die maximale Länge eines 

Blattnamens ist. Länger kann er nicht sein. In unserem Beispiel sind es nur 6 Zeichen. 

Wir schreiben trotzdem 31 hin. Das macht nichts, denn wenn es weniger Zeichen 

sind, schneidet Excel eben so viele Zeichen heraus, wie da sind – u.U. weniger als 31. 

ACHTUNG: Das zweite Argument von TEIL(…) gibt an, ab welcher Position etwas 

herausgeschnitten werden soll. Das dritte Argument gibt nicht an, bis zu welcher Posi-

tion, sondern wie viele Zeichen herausgeschnitten werden sollen. Das wird gerne 

falsch gemacht! 

Das Ergebnis der o.g. TEIL(…)-Formel ist dann also in unserem Beispiel: 1-2019. 

Genau das sehen Sie in Zelle A2. 

15  
Jetzt wollen wir aus dem Inhalt von A2 wieder etwas herausschneiden: Die Nummer 

des Monats. Darum schreiben wir in Zelle C2 die Formel: 

=LINKS(A2;LÄNGE(A2)-5) 

Sie schneidet von links beginnend (darum heißt sie LINKS(…) ;-) einige Zeichen aus 

dem Inhalt von A2 heraus. Wie viele? Das gibt das zweite Funktionsargument an: 

LÄNGE(A2)-5 

LÄNGE(…) ermittelt die Gesamtanzahl von Zeichen in A2. Das sind in unserem Bei-

spiel 6 („1-2019“). Fünf davon sind die Jahreszahl und das Minuszeichen. Also ziehen 

wir diese 5 von der Gesamtzahl der Zeichen ab. LÄNGE(A2)-5 schneidet uns also 

jetzt 6-5=1 Zeichen von links aus A2 heraus: „1“! 

16  
Jetzt müssen wir noch den Mitarbeiternamen aus A5 herausschneiden. Er befindet 

sich zwischen den beiden eckigen Klammern. Wir müssen also zunächst herausfin-

den, an wievielter Position sich diese jeweils befinden. 

Dafür schreiben Sie die Formeln  

in E2: =FINDEN("[";A5) 

in F2: =FINDEN("]";A5) 

Daraufhin erscheint in E2 die Zahl 37 und in F2 die Zahl 64. Die öffnende eckige 

Klammer befindet sich also an 37. Position in A5 und die schließende an 64. Position 

innerhalb des Textes in Zelle A5. (Bei Ihnen werden das natürlich andere Zahlen 

sein.) 

Dazwischen befindet sich der Mitarbeitername. 

ACHTUNG: Die Formel FINDEN(…) liefert als Ergebnis eine Zahl – keinen Text! 

17  
Schreiben Sie jetzt die Formel 

=TEIL(A5;E2+6;F2-E2-19) 

in Zelle D4. Daraufhin erscheint dort der Mitarbeitername. 

TEIL(…) schneidet aus A5 ab der Position E2+6 eine bestimmte Anzahl von Zeichen 

aus – und zwar F2-E2-19 Stück. 

Überlegen Sie mal selbst, warum es E2+6 bzw. F2-E2-19 heißen muss! 

Für eine detaillierte Erläuterung der Formel TEIL(…) siehe Pkt. 14! 

18  
Jetzt brauchen wir noch den Monat und das Jahr in H4 und J4. Dafür schreiben Sie 

folgende Formeln 

in H4: =TEXT(DATUM(2019;C2;1);"MMMM") 

in J4: =RECHTS(A2;4) 

DATUM(…) erzeugt aus den drei Bestandteilen Jahr („2019“), Monat (C2) und Tag („1“) 

ein Datum. TEXT(…) legt das Format fest, in dem dieses angezeigt werden soll. 

„MMMM“ bedeutet: „Der voll ausgeschriebene Monatsname.“ 

Um zu erkennen, wie das funktioniert, probieren Sie auch mal „M“, „MM“ und „MMM“! 

19  
Ihr Blatt sieht jetzt so ähnlich aus: 

„Ähnlich“, weil Sie Ihre Datei sicher in einem anderen Ordner abgespeichert haben, 

d.h. Sie haben in A5, E2 und F2 etwas Anderes stehen. Der Rest müsste bei Ihnen 

genauso aussehen wie in der obigen Abbildung.  

20  
Die Inhalte in den Zeilen 2 und 5 sind nur Hilfswerte, die wir zur Ermittlung des Mitar-

beiternamens in D4 und des Monats und des Jahres in H4 und J4 benötigen. Sie sol-

len unsichtbar sein. Daher formatieren Sie die Zeilen 2 und 5 so: 

Start / Schriftart / Schriftfarbe = weiß 

 

Dadurch verschwinden diese Hilfswerte (weiße Schrift auf weißem Hintergrund!). Jetzt 

besteht natürlich die Gefahr, dass der unerfahrene Benutzer des Stundenzettels den 

Inhalt dieser unsichtbaren Zellen durch Überschreiben mit irgendwelchen anderen Da-

ten zerstört. Darum schalten wir am Ende (Pkt. 31) den Blattschutz ein, der so etwas 

verhindert. 

21  
Jetzt wollen wir in Spalte A den Wochentag und in Spalte B das Datum haben. Dafür 

schreiben Sie folgende Formeln 

in B8: =DATUM(J4;C2;1) 

in B9: =B8+1 

in A8: =TEXT(WOCHENTAG(B8);"TTT")&"." 

Anschließend kopieren Sie die Formel in B9 herunter bis B38 und die Formel in A8 

herunter bis A38.  

„Herunter kopieren“ bedeutet: Klicken Sie mit der linken Maustaste auf den kleinen 

Punkt in der rechten unteren Ecke der Zelle, halten Sie die Maustaste gedrückt und 

ziehen Sie den Zeiger nach unten: 

In der Spalte „Datum“ wollen wir die Jahreszahl nicht sehen. Dort soll z.B. nur stehen 

„01.02.“. Dafür benötigen wir eine spezielle Formatierung: 

Markieren Sie den Bereich B8:B38 und wählen Sie dann 

Start / Zahl / Mehr … / Benutzerdefiniert 

Als Typ geben Sie ein: TT.MM. 

TEXT(…) formatiert wieder den Wochentag mittels der Angabe „TTT“. Probieren Sie 

auch mal „T“, „TT“ und „TTTT“ aus! 

22  
Die Zeilen, die ein Wochenende enthalten (also „Sa.“ oder „So.“ in Spalte A) sollen 

farblich hervorgehoben werden. Das machen wir natürlich nicht von Hand, sondern 

lassen es Excel wieder automatisch erledigen! Dazu benutzen wir die sog. „Bedingte 

Formatierung“. Damit ist es möglich, festzulegen, dass bestimmte Zellen in Abhängig-

keit vom Inhalt anderer Zellen in bestimmter Weise formatiert werden. 

Ich kann also z.B. sagen: Wenn in Spalte A einer Zeile „Sa.“ steht, färbe die Zellen 

diese Zeile grau. Das richten Sie folgendermaßen ein: 

Markieren Sie den Bereich A8:J38, wählen Sie 

Start / Formatvorlagen / Bedingte Formatierung / Neue Regel 

ACHTUNG – auf dem Mac: Start / Bedingte Formatierung / Regeln zum Hervorheben 

von Zellen / Weitere Regeln 

In dem Fenster, das sich daraufhin öffnet, wählen Sie unter Regeltyp auswählen den 

Eintrag Formel zur Ermittlung der zu formatierenden Zelle verwenden. 

Dort tragen Sie folgende Formel ein: 

=ODER($A8="Sa.";$A8="So.") 

Das sieht dann so aus: 

Jetzt klicken Sie rechts unten in dem Fenster auf Formatieren. In dem Fenster, das 

sich daraufhin öffnet, wählen Sie im Register Ausfüllen einen grauen Hintergrund aus 

und klicken auf OK. 

Wenn Sie jetzt im Fenster Neue Formatierungsregel nochmal auf OK klicken, werden 

die Wochenenden wie gewünscht markiert. 

 

Erläuterung der Formel: 

Erstens: Es handelt sich hier um die Verknüpfung zweier Bedingungen. Die Formel ist 

so zu lesen: 

WENN $A8=“Sa:“ ODER $A8=“So.“ 

Das ODER steht nur nicht – wie im normalen Sprachgebrauch – zwischen den beiden 

Alternativen, sondern davor. 

 

Zweitens: In der Formel sehen Sie ein Dollarzeichen als Bestandteil der Zelladresse: 

„$A8“. Das nennt man „absolute Adressierung“ und hat folgende Auswirkung: 

Wenn Sie eine Formel waagerecht oder senkrecht kopieren, werden ja bekanntlich die 

darin enthaltenen Zelladressen automatisch verändert. Beim waagerechten Kopieren 

wird aus A8: B8, C8, … usw. Beim senkrechten Kopieren wird aus A8: A9, A10, … 

usw. Das Dollarzeichen verhindert diese Anpassung. Beim waagerechten Kopieren 

wird dann aus $A8: $A8, $A8, … usw., d.h. die Zelladresse bleibt unverändert. Beim 

senkrechten Kopieren wird dann aber nach wie vor aus $A8: $A9, $A10, … usw. 

So ist das beim Kopieren von Formeln. Was hat das jetzt aber mit der bedingten For-

matierung zu tun? Da wird ja nichts kopiert! 

Doch – indirekt schon! Wir hatten ja den Bereich A8:J38 markiert, bevor wir die be-

dingte Formatierung eingeschaltet haben. In diesem Bereich soll die bedingte Forma-

tierung ausgeführt werden, d.h. wenn in Spalte A „Sa.“ oder „So.“ steht, soll nicht nur 

die Zelle in Spalte A grau werden, sondern die ganze Zeile bis Spalte J. 

Die bedingte Formatierung wird jetzt so ausgeführt, als hätte man die Formel 

=ODER($A8="Sa.";$A8="So.") 

in der linken oberen Ecke des markierten Bereichs (also in A8) eingegeben und dann 

nach unten (bis A38) und rechts (bis J38) kopiert! 

Hätten wir die Formel ohne Dollarzeichen eingegeben, würde die Formel für Zelle B8 

so lauten: 

=ODER(B8="Sa.";B8="So.") 

und für C8 so: 

=ODER(C8="Sa.";C8="So.") 

In B8 bzw. C8 steht aber weder „Sa.“ noch „So.“. Folglich wird die Zelle B8 bzw. C8 

auch nicht grau! Probieren Sie das bitte mal aus, indem Sie die Formel in der beding-

ten Formatierung entsprechend ändern! Sie werden sehen, dass dann nur die Zelle in 

Spalte A grau wird – und nicht wie gewünscht die ganze Zeile! 

Ich weiß – die bedingte Formatierung ist relativ schwer zu verstehen – aber wie Sie 

sehen, man kann damit schicke Sachen machen. Es lohnt sich also, sich die Mühe zu 

machen, das zu verstehen! 

ACHTUNG: Sie können das Dollarzeichen mit der entsprechenden Taste eingeben – 

Sie können aber auch die Zelladresse, die ein Dollarzeichen erhalten soll, mit der lin-

ken Maustaste anklicken und dann die Taste F4 drücken. Diese hat bei mehrfachem 

Drücken vier Einstellungen: B8, $B$8, $B8 und B$8. Diese Methode ist sehr zu emp-

fehlen, denn damit vermeiden Sie Fehleingaben! Bitte ausprobieren! 

23  
ACHTUNG: Damit Sie für die Spaltenbreite die Maßeinheit cm benutzen können, sind 

zwei Einstellungen vorzunehmen: 

(1) Datei / Optionen / Erweitert / Anzeige / Linealeinheiten: Zentimeter 

Mac: Ganz oben im Menü neben dem Apfel auf Excel / Einstellungen / Allgemein 

/ Linealeinheiten 

(2) Ansicht / Arbeitsmappenansichten / Seitenlayout 

Richten Sie jetzt die Spaltenbreiten folgendermaßen ein: 

A:  1cm 

B-G: 1,6cm 

H:  0,7cm 

I:  0,9cm 

J:  so breit, dass sie noch auf die Seite passt 

Dazu klicken Sie mit der rechten Maustaste auf den Spaltenbuchstaben (z.B. A) und 

wählen aus dem Kontextmenü Spaltenbreite. Wenn Sie mehrere Spalten gleichzeitig 

formatieren wollen (z.B. B bis G), markieren Sie diese zunächst. Dazu klicken Sie mit 

der linken Maustaste auf den ersten Spaltenbuchstaben (z.B. B) und bewegen dann 

den Mauszeiger mit gedrückter linker Taste bis zum letzten Spaltenbuchstaben (z.B. 

G). Jetzt klicken Sie mit der rechten Maustaste auf irgendeinen der markierten Spal-

tenbuchstaben und wählen aus dem Kontextmenü Spaltenbreite.  

ACHTUNG: Sollte anschließend eine Spalte so aussehen: 

 

… dann ist sie zu schmal, um die darin enthaltenen Daten anzuzeigen. Wenn Sie sie 

etwas breiter machen, verschwinden die Rauten wieder. 

24  
Ihr Blatt sieht jetzt so aus: 

 

25  
Die Überschrift „F“ in Spalte H bedeutet „Fahrer“. Hier soll ein Kreuz gemacht werden, 

wenn der Mitarbeiter an diesem Tag das Firmenfahrzeug gefahren hat. 

In Spalte I kann eingetragen werden 

U = Urlaub 

K = krank 

F = frei  

FT = Feiertag 

Wir wollen es dem Benutzer möglichst einfach machen, diese Werte einzutragen. 

Gleichzeitig wollen wir aber auch verhindern, dass er dort falsche Werte einträgt – 

also z.B. ein X in Spalte I. Dafür benutzen wir die Excel-Funktion „Datenüberprüfung“. 

Markieren Sie den Bereich H8:H38 und klicken Sie auf 

Daten / Datentools / Datenüberprüfung 

Dort wählen Sie unter Zulassen die Option Liste. Dann tragen Sie unter Quelle ein X 

ein und klicken auf OK. 

Wenn Sie jetzt in eine Zelle in Spalte H klicken, erscheint rechts daneben ein Button, 

mit dem Sie eine Combobox öffnen können. Sie enthält nur einen Wert: X. Diesen 

können Sie mit der Maus auswählen. Sie können das X aber auch über die Tastatur 

eingeben. Versuchen Sie aber mal, einen anderen Wert einzugeben … 

Entsprechend gehen wir für Spalte I vor. Markieren Sie den Bereich I8:I38 und klicken 

Sie auf 

Daten / Datentools / Datenüberprüfung 

Dort wählen Sie unter Zulassen die Option Liste. Dann tragen Sie unter Quelle ein:  

B;U;K;F;FT und klicken auf OK. 

Wenn Sie jetzt in eine Zelle in Spalte I klicken, erscheint rechts daneben ein Button, 

mit dem Sie eine Combobox öffnen können. Sie enthält die Werte B, U, K, F und FT. 

Davon können Sie einen mit der Maus auswählen. Sie können einen dieser Werte 

aber auch über die Tastatur eingeben. Versuchen Sie aber mal, einen anderen Wert 

einzugeben … 

26  
Jetzt sind wir fast fertig. Es fehlen nur noch die Berechnung der täglichen Arbeitszeit 

in Spalte F und die Summen der Spalten E, F und G. Dafür formatieren Sie den Be-

reich unterhalb der Monatstabelle entsprechend der Abbildung auf Seite 2 dieser An-

leitung und tragen Sie folgende Formeln ein: 

in E40: =SUMME(E8:E38)/60 (=Gesamt-Pausenzeit in Stunden) 

in F40: =SUMME(F8:F38) (=Gesamt-Arbeitszeit in Stunden) 

in G40: =SUMME(G8:G38)/60 (=Gesamt-Fahrzeit in Stunden) 

in F42: =F40-E40-G40 (=Gesamt-Arbeitszeit minus Gesamt-Pausenzeit 

 minus Gesamt-Fahrzeit) 

Die Angabe „E8:E38“ bezeichnet einen Zellbereich. Der erste Wert (E8) ist die linke 

obere Ecke des Bereiches, der zweite Wert (E38) ist die rechte untere Ecke. Die so 

bezeichneten Zellen müssen nicht unbedingt untereinander in einer Spalte oder ne-

beneinander in einer Zeile liegen. Sie könnten auch mit SUMME(E8:G38) alle Zahlen 

in einem rechteckigen Bereich aus mehreren Zeilen und Spalten summieren. 

ACHTUNG: Ein beliebter Fehler besteht darin, zwischen die beiden Zelladressen ein 

Semikolon statt eines Doppelpunktes zu machen. Das funktioniert (leider!) auch, be-

deutet aber etwas Anderes. Das ist nämlich kein Zellbereich, sondern eine Aufzählung 

einzelner Zellen. SUMME(E3;E38) bedeutet also, dass nur die Inhalte der beiden Zel-

len E8 und E38 addiert werden statt des ganzen Bereiches von E8 bis E38. 

In F8 tragen Sie die Formel 

=WENN(UND(C8<>"";D8<>"");(D8-C8)*24;"") 

ein und kopieren sie bis F38 herunter. Sie bewirkt folgendes: 

Wenn in den Zellen C8 und D8 etwas steht („etwas“ = „ungleich nichts“ = <>““!), dann 

wird die Arbeitszeit aus der Differenz zwischen Ende und Beginn berechnet, sonst 

wird nichts ("") in die Zelle geschrieben. 

Für UND(…) gilt wieder, was weiter oben schon für ODER(…) gesagt wurde: 

Es handelt sich hier um die Verknüpfung zweier Bedingungen. Die Formel ist so zu le-

sen: 

WENN C8<>““ UND D8<>““ DANN … 

Das UND steht nur nicht – wie im normalen Sprachgebrauch – zwischen den beiden 

Alternativen, sondern davor. 

Warum wird die Differenz mit 24 multipliziert? Die Differenz zweier Zeitangaben hat 

die Maßeinheit „Tage“! Um Stunden anzuzeigen, müssen Sie also mit 24 multiplizie-

ren. 

27  
Formatieren Sie den Bereich C8:D38 auf Benutzerdefiniert: hh:mm:  

Start / Zahl / Mehr … / Benutzerdefiniert 

Jetzt können Sie in den Spalten C, D, E, G, H und I testweise einige Werte eingeben 

und beobachten, wie die Werte in Spalte F und die Summen unterhalb der Tabelle be-

rechnet werden. 

28  
Fertig? Leider noch nicht :-( 

Ändern Sie mal den Namen des Blattes auf 2-2019. Wenn Sie bis hierhin alles richtig 

gemacht haben, funktioniert unsere Automatisierungstechnik wunderbar: 

In H4 erscheint „Februar“, in den Spalten A und B erscheinen die Angaben für Februar 

und auch die Wochenenden werden automatisch richtig markiert. 

Wunderbar! 

ABER: Wenn Sie sich mal das untere Ende der Tabelle ansehen, dann gibt es dort 

jetzt Zeilen mit Datumsangaben aus dem März. Klar – denn wir haben ja 31 Zeilen an-

gelegt! Das müssen wir verhindern! 

Dafür benutzen wir wieder die bedingte Formatierung, indem wir fordern: 

„Wenn in den Zeilen 35 bis 38 der Tag kleiner als 5 ist, sollen diese Zeilen nicht ange-

zeigt werden!“ Dabei bedeutet „nicht angezeigt“ = „weiße Schrift auf weißem Hinter-

grund“! 

Markieren Sie den Bereich A35:J38 und wählen Sie 

Start / Formatvorlagen / Bedingte Formatierung / Regel verwalten / Neue Regel 

Regeltyp ist wieder Formel zur Ermittlung der zu formatierenden Zellen verwenden. 

Die Regel lautet: 

=TAG($B35)<5 

Formatierung: Schriftfarbe = weiß 

ACHTUNG: Wichtig ist jetzt noch, dass Sie das Häkchen bei Anhalten richtig setzen: 

 

Dieses bewirkt, dass nach dem Zutreffen der ersten Regel die zweite Regel nicht 

mehr angewendet wird. Das bedeutet: Wenn die Tage nicht angezeigt werden sollen 

(=weiße Schrift!), dann sollen auch die Wochenenden nicht markiert werden. Das 

wäre ja sinnlos. 

Sie können die Wirkungsweise des Häkchens bei Anhalten überprüfen, indem Sie es 

probehalber entfernen und sich dann mal die letzten Tage auf dem Blatt 2-2019 anse-

hen. Fällt Ihnen etwas auf? 

Beachten Sie auch hier wieder das Dollarzeichen in der Zelladresse $B35! (siehe Er-

läuterung dazu unter Pkt. 22) 

29  
Jetzt sind wir wirklich fertig! 

Sie können jetzt Blätter für alle Monate des Jahres anlegen, die Datei speichern und 

schließen und dann für mehrere Mitarbeiter kopieren. 

30  
Stopp – es geht doch noch weiter! 

Sie haben die fertige Excel-Datei beim Auftraggeber vorgestellt – dabei kommt her-

aus: Die Fahrzeit wird noch falsch berücksichtigt. Es soll nämlich so sein: 

Die in Spalte G eingetragene Fahrzeit setzt sich aus der Summe für Hin- und Rück-

fahrt zusammen. Dabei wird angenommen, dass beide gleich lange dauern. Eine 40 in 

Spalte G bedeutet also: 20 Minuten für die Hinfahrt und 20 für die Rückfahrt. 

Diese Fahrzeit ist in der Arbeitszeit (Spalte F) bereits enthalten. Es gilt also „Beginn“ = 

„Vor Antritt der Hinfahrt“ und „Ende“ = „Nach Ende der Rückfahrt“. 

Beim Fahrer (=Kreuz in Spalte H) soll die gesamte Fahrzeit (Hin- plus Rückfahrt) als 

Arbeitszeit zählen, beim Beifahrer (=kein Kreuz in Spalte H) nur die halbe. 

Für den Beifahrer (=kein Kreuz in Spalte H) müssen wir also jetzt die halbe Fahrzeit 

von der Arbeitszeit abziehen. 

Das machen wir folgendermaßen: 

Wir fügen zwischen Spalte G und H eine neue Spalte ein und berechnen dort die je-

weils abzuziehende Fahrzeit in Abhängigkeit davon, ob rechts daneben ein X steht o-

der nicht. 

Also: 

Markieren Sie die Spalte H, indem Sie auf den Buchstaben H klicken. Dann wählen 

Sie Start / Zellen / Einfügen. Links neben Spalte H erscheint eine neue Spalte, die 

jetzt H heißt. Die ehemalige Spalte H heißt jetzt I, … usw. 

Schreiben Sie in H8 folgende Formel: 

=WENN(I8="X";0;G8/2) 

und ziehen Sie sie bis H38 herunter. Die Formel bewirkt folgendes: Wenn rechts da-

neben ein X steht (=Fahrer), wird nichts abgezogen, sonst (=Beifahrer) die halbe 

Fahrzeit aus Spalte G. 

Die Spalte H soll aber gar nicht sichtbar sein. Sie wird nur für die Berechnung der 

Fahrzeiten benötigt. Wir blenden sie also aus. Klicken Sie dazu mit der rechten Maus-

taste auf den Buchstaben H und wählen Sie aus dem sich öffnenden Kontextmenü 

den Eintrag Ausblenden. 

ACHTUNG: Jetzt müssen wir auch die Formel in G40 ändern, die uns die Summe der 

Fahrzeiten anzeigt. Dort müssen wir nicht mehr Spalte G, sondern die ausgeblendete 

Spalte H aufsummieren: 

=SUMME(H8:H38)/60  

Sie können eine ausgeblendete Spalte (z.B. H) wieder einblenden, indem Sie mit der 

linken Maustaste einen Spaltenbuchstaben links davon (z.B. G) anklicken und dann den 

Mauszeiger mit gedrückter linker Maustaste auf einen Spaltenbuchstaben rechts von 

der ausgeblendeten Spalte (z.B. I) ziehen. Jetzt haben Sie die ausgeblendete Spalte 

mit markiert (obwohl sie unsichtbar ist) und können mit der rechten Maustaste auf einen 

der markierten Spaltenbuchstaben klicken und im Kontextmenü einblenden wählen! 

31  
Damit der Benutzer Zellen mit Formeln darin nicht überschreiben kann, sollten Sie den 

Blattschutz einschalten. Die Logik dabei ist folgende: 

Sie schalten den Schutz für einzelne Zellen AUS und dann den Schutz für das ge-

samte Blatt EIN. Dann bleiben die Zellen ungeschützt, die Sie vorher vom Schutz aus-

genommen haben. In diese Zellen kann der Benutzer etwas hineinschreiben. 

Gehen Sie folgendermaßen vor:  

 Klicken Sie in die Zelle D3, 

 drücken Sie die Strg-Taste und halten Sie diese gedrückt. 

 Markieren Sie dann die beiden Bereiche C8:E38 und G8:K38 

 Lassen Sie die Strg-Taste los. 

 Klicken Sie mit der rechten Maustaste in den markierten Bereich und wählen 
Sie Zellen formatieren. 

 In dem Fenster Zellen formatieren, das sich daraufhin öffnet, entfernen Sie auf 
dem Register Schutz das Häkchen bei Gesperrt. 

 Klicken Sie auf OK. 

 Wählen Sie jetzt Überprüfen / Änderungen / Blatt schützen 

 Jetzt müssen Sie entscheiden, ob Sie ein Kennwort eingeben wollen. 
ACHTUNG: Wenn Sie es vergessen, kann Ihnen nichts und niemand mehr hel-
fen, den Blattschutz wieder aufzuheben. Am besten, erst mal kein Kennwort 
eingeben. 

 Klicken Sie auf OK. 

Jetzt kann der Benutzer nur noch in D3, C8:E38 und G8:K38 Daten eingeben. Alle an-

deren Zellen sind gegen Veränderungen ihrer Inhalte geschützt. 

ACHTUNG: Bei eingeschaltetem Blattschutz sind viele Excel-Funktionen in den Me-

nüs nicht benutzbar! Im Menü Start sind z.B. alle Formatierungsfunktionen deaktiviert: 

 

Wenn Sie also auf einem Excel-Blatt bestimmte Funktionen nicht benutzen können, 

müssen Sie den Blattschutz ausschalten: 

Überprüfen / Änderungen / Blattschutz aufheben 

32  
Testen Sie jetzt die Funktion Ihres Arbeitsergebnisses, indem Sie dieselben Aktionen 

durchführen wie ganz zu Anfang unter Pkt. 1! 

Überprüfen Sie insbesondere, ob die Berechnungen richtig durchgeführt werden, in-

dem Sie die Aufsummierung der Arbeitszeiten im Kopf oder mit dem Taschenrechner 

nachrechnen. Wird auf das X in Spalte I richtig reagiert? 

Testen, testen, testen!! 

Nicht glauben „Excel hat‘s gemacht – wird schon stimmen!“ 

Beim Testen müssen Sie immer überlegen, welche Werte der Benutzer eingeben 

kann – hier also für Beginn, Ende, Pause und Fahrzeit. Funktioniert Ihr Blatt wirklich 

für ALLE DENKBAREN Werte? Wird verhindert, dass der Benutzer falsche oder un-

sinnige Werte eingibt? Wie reagiert Ihr Blatt in einem solchen Fall? 

Denksportaufgabe: 

Im Falle „ganz besonderer Arbeitszeiten“ wird noch etwas falsch gemacht. Finden Sie 

es heraus? 

… und das haben Sie jetzt gelernt: 

Wo? Was? 

1 Blatt kopieren und umbenennen 

2 Blatt umbenennen, Seitenränder und Seitenansicht einstellen 

3 gesamtes Blatt markieren, Schriftart einstellen 

4 Zellen verbinden und formatieren 

5 adressieren zusammengefasster Zellen 

11 Rahmenlinien definieren 

Zeilenumbruch in einer Zelle mit Alt-Enter 

13 Ermittlung des Datei- bzw. Blattnamens mit der Formel ZELLE(…) 

14 Textbearbeitung mit den Formeln TEIL(…) und FINDEN(…) 

15 Textbearbeitung mit den Formeln LINKS(…) und LÄNGE(…) 

16 Textbearbeitung mit der Formel FINDEN(…) 

18 Formatierung mit den Formeln TEXT(…) und DATUM(…)  

Textbearbeitung mit der Formel RECHTS(…) 

20 Zellinhalte unsichtbar machen durch weiße Schrift auf weißem Hintergrund 

21 Ermittlung des Wochentages mit der Formel WOCHENTAG(…) 

Eine Formel herunter kopieren 

Formatierungen „T“, „TT“, „TTT“ und „TTTT“ sowie „M“, „MM“, „MMM“ und „MMMM“ 

22 Bedingte Formatierung mit einer Formel als Bedingung 

Absolute Zell-Adressierung mit Dollarzeichen; Eingabe mit der F4-Taste 

Verknüpfung zweier Alternativen mit der Formel ODER(…) 

23 Maßeinheit Zentimeter für die Spaltenbreite und Zeilenhöhe einstellen 

Einstellung der Spaltenbreite; Problem „nur Rauten in der Spalte“ beheben 

25 Einsatz der Datenüberprüfung zur schnellen und sicheren Dateneingabe (dort insbesondere 

die Option Liste) 

26 Formeln SUMME(…), WENN(…) und UND(…) 

Bezeichnung eines Zellbereiches durch „linke obere Ecke : rechte untere Ecke“ 

„wenn in der Zelle etwas steht …“ (<>““) 

Umrechnung von Zeitdifferenzen in Stunden 

27 Benutzerdefiniertes Format („hh:mm“) einstellen 

28 Formel TAG(…) 

Ausführung der bedingten Formatierung nach Anwendung einer Regel anhalten 

30 Spalte einfügen bzw. aus- und einblenden 

31 mehrere, auch nicht zusammenhängende Zellbereiche mit gedrückter Strg-Taste markieren 

Blattschutz einschalten / ausschalten 

… und wenn Sie das alles wirklich gelernt haben, müssten Sie jetzt diese Fragen be-
antworten können: (OHNE weiter oben nachzusehen – damit würden Sie sich ja selber 
betrügen …) 
 

1.  Wie kopiert man ein Excel-Blatt und benennt es um? 

2.  Wie markiert man mit einem Klick alle Zellen eines Arbeitsblattes? 

3.  Wie sind die verbundenen Zellen nach der Ausführung von „Verbinden und zentrie-

ren“ formatiert? Wie muss diese Formatierung ggf. geändert werden? 

4.  Wie werden zusammengefasste Zellen in einer Formel adressiert? 

5.  Wie erreicht man einen Zeilenumbruch innerhalb einer Zelle? 

6.  Welches Ergebnis liefert die Formel ZELLE(„dateiname“; $A$1)? 

7.  Zur Formel TEIL(Text;Erstes_Zeichen;Anzahl_Zeichen): 

 Welche Argumente hat die Formel? 

 Welches Format haben diese Argumente? (Text, Zahl, Datum, Bereich, …) 

 Welches Ergebnis liefert die Formel? 

 Welches Format hat das Ergebnis? (Text, Zahl, Datum, …) 

8.  Zur Formel FINDEN(Suchtext;Text;[Erstes_Zeichen]): 

 Welche Argumente hat die Formel? 

 Welches Format haben diese Argumente? (Text, Zahl, Datum, Bereich, …) 

 Welches Ergebnis liefert die Formel? 

 Welches Format hat das Ergebnis? (Text, Zahl, Datum, …) 

9.  Zur Formel LINKS(Text;[Anzahl_Zeichen]): 

 Welche Argumente hat die Formel? 

 Welches Format haben diese Argumente? (Text, Zahl, Datum, Bereich, …) 

 Welches Ergebnis liefert die Formel? 

 Welches Format hat das Ergebnis? (Text, Zahl, Datum, …) 

10.  Zur Formel RECHTS(Text;[Anzahl_Zeichen]): 

 Welche Argumente hat die Formel? 

 Welches Format haben diese Argumente? (Text, Zahl, Datum, Bereich, …) 

 Welches Ergebnis liefert die Formel? 

 Welches Format hat das Ergebnis? (Text, Zahl, Datum, …) 

11.  Zur Formel LÄNGE(Text): 

 Welche Argumente hat die Formel? 

 Welches Format haben diese Argumente? (Text, Zahl, Datum, Bereich, …) 

 Welches Ergebnis liefert die Formel? 

 Welches Format hat das Ergebnis? (Text, Zahl, Datum, …) 

12.  Zur Formel TEXT(Wert;Formatcode): 

 Welche Argumente hat die Formel? 

 Welches Format haben diese Argumente? (Text, Zahl, Datum, Bereich, …) 

 Welches Ergebnis liefert die Formel? 



Excel-Übung “Stundenzettel” www.andreasstern.de 

Stundenzettel-V190131.docx  Seite 20 von 21 

 Welches Format hat das Ergebnis? (Text, Zahl, Datum, …) 

13.  Was bewirken die Formatangaben 

 „T“, „TT“, „TTT“, „TTTT“ für Tage 

 „M“, „MM“, „MMM“, „MMMM“ für Monate? 

14.  Zur Formel DATUM(Jahr;Monat;Tag): 

 Welche Argumente hat die Formel? 

 Welches Format haben diese Argumente? (Text, Zahl, Datum, Bereich, …) 

 Welches Ergebnis liefert die Formel? 

 Welches Format hat das Ergebnis? (Text, Zahl, Datum, …) 

15.  Wie kopiert man eine Formel in alle untereinander liegenden Zellen einer Spalte bzw. 

alle nebeneinander liegenden Zellen einer Zeile? 

16.  Wie weist man einer Zelle ein benutzerdefiniertes Format zu? 

17.  Wie kann man den Inhalt einer Zelle unsichtbar machen? 

18.  Wie funktioniert die bedingte Formatierung? 

19.  Zur Formel ODER(Bedingung; Bedingung; [Bedingung];…): 

 Welche Argumente hat die Formel? 

 Welches Format haben diese Argumente? (Text, Zahl, Datum, Bereich, …) 

 Welches Ergebnis liefert die Formel? 

 Welches Format hat das Ergebnis? (Text, Zahl, Datum, …) 

20.  Zur Formel UND(Bedingung; Bedingung; [Bedingung];…): 

 Welche Argumente hat die Formel? 

 Welches Format haben diese Argumente? (Text, Zahl, Datum, Bereich, …) 

 Welches Ergebnis liefert die Formel? 

 Welches Format hat das Ergebnis? (Text, Zahl, Datum, …) 

21.  Was bedeutet „absolute Zell-Adressierung“? Wie muss die entsprechende Zella-

dresse eingegeben werden?  

22.  Welche besondere Methode gibt es, um zu einer Zelladresse Dollarzeichen hinzu-

zufügen (=Umwandlung in eine absolute Adresse)? 

23.  Wie können Sie die Breite mehrerer Spalten gleichzeitig einstellen? 

24.  Was bedeutet es, wenn in einer Spalte nur Rauten (#) zu sehen sind? Was können 

Sie tun? 

25.  Wie funktioniert die „Datenüberprüfung“ – insbesondere mit der Option „Liste“? 

26.  Zur Formel WENN(Bedingung;Dann-Wert;Sonst-Wert): 

 Welche Argumente hat die Formel? 

 Welches Format haben diese Argumente? (Text, Zahl, Datum, Bereich, …) 

 Welches Ergebnis liefert die Formel? 

 Welches Format hat das Ergebnis? (Text, Zahl, Datum, …) 

27.  Wie wird in einer Formel ein Zellbereich angegeben? 

28.  Worin besteht der Unterschied zwischen SUMME(E8:E38) und SUMME(E8;E38)? 

29.  Wie schreiben Sie in einer Formel „Wenn in der Zelle A1 etwas steht“?  

30.  Welche Maßeinheit hat die Differenz zweier Zeitangaben? 

31.  Wie rechnen Sie die Differenz zweier Zeitangaben in Stunden um? 

32.  Was bedeutet das Häkchen bei „Anhalten“ bei der bedingten Formatierung? 

33.  Wie werden mehrere (auch nicht zusammenhängende) Zellbereiche markiert? 

 


